1. Algebraiska räkningar

Repetera **addition**, **multplikation** osv. (datoranvändning). Observera att *maple* räknar exakt, om du inte anger annat; *evalf* räknar ut ett numeriskt närmevärde (floating-point number), det får du också om du skriver minst ett tal som reellt tal, t.ex. 2.0 i.st.f. 2:

Exempel 1

```
> 2/3+4/5;
evalf(%);
2.0/3+4/5;
```

$$\frac{22}{15}$$

$$1.466666667$$

$$1.466666667$$

Faktorerna till ett heltal får du med *ifactor* (integer-faktorisering):

Exempel 2

```
> ifactor(123456789);
```

$$(3)^2\ (3803)\ (3607)$$

Du kan faktoruppdela polynom med *factor*, om det har rationella rötter:

Exempel 3

```
> p1:=x^4-19/6*x^3+8/3*x^2-19/6*x+5/3;
```

$$p1 := x^4 - \frac{19}{6}x^3 + \frac{8}{3}x^2 - \frac{19}{6}x + \frac{5}{3}$$

```
> factor(p1);
```

$$\frac{(2\,x-5)\,(3\,x-2)\,(x^2+1)}{6}$$

Omvänt, du kan "*utveckla*" (räkna ut) produkter med *expand*; pröva även *normal*:

Exempel 4

```
> p2:=(sqrt(2*x)-x/3)^4;
```

$$p2 := \left(\sqrt{2}\,\sqrt{x} - \frac{x}{3}\right)^4$$

```
> expand(p2);
```

$$4\,x^2 - \frac{8\sqrt{2}\,x^{(5/2)}}{3} + \frac{4\,x^3}{3} - \frac{4\sqrt{2}\,x^{(7/2)}}{27} + \frac{x^4}{81}$$

```
> p3:=normal((x-2)^3+x^3-(2*x-3)^2);
```

$$p3 := 2\,x^3 - 10\,x^2 + 24\,x - 17$$

Lösningar till **ekvationer** " $f(x) = g(x)$ " får du med *solve* resp. *fsolve* (numeriskt); skriver du bara $f(x)$ så löses $f(x) = 0$; hur *maple* svarar styrs av hur du anger variabeln ("sök x så att..."), behöves ej om det bara finns en variabel; kolla följande

Exempel 5

```
> solve(p1);solve(p1,{x});
```

$$\frac{2}{3}, \frac{5}{2}, I, -I$$

$$\{x = \frac{2}{3}\}, \{x = \frac{5}{2}\}, \{x = I\}, \{x = -I\}$$

```
> solve(p2);fsolve(p3);
```

$$0, 18$$

$$1.104109859$$

```
> solve(a*x^2+b*x+c,x);
```

$$\frac{-b + \sqrt{b^2 - 4\,a\,c}}{2\,a}, \frac{-b - \sqrt{b^2 - 4\,a\,c}}{2\,a}$$

Analogt för flera ekvationer för flera obekanta:

Exempel 6

```
> ekv1:=5*x^2-3*y^2=2; ekv2:=3*x-2*y=3;
solve({ekv1,ekv2},{x,y});
```

$$ekv1 := 5\,x^2 - 3\,y^2 = 2$$

$$ekv2 := 3\,x - 2\,y = 3$$

$$\{x = \frac{5}{7}, y = \frac{-3}{7}\}, \{x = 7, y = 9\}$$

Alla lösningar (även komplexa) får du ev. så här (du måste ange variabeln):

Exempel 7

```
> fsolve(p3,{x},complex);
```
(p3 från ex.4)

$$\{x = 1.104109859\}, \{x = 1.947945071 - 1.975859127\,I\},$$
$$\{x = 1.947945071 + 1.975859127\,I\}$$

Olikheter löses på samma sätt; observera hur *maple* svarar beroende på om du anger variabeln eller ej (se även ex. 15):

Exempel 8

```
> olikhet:=(x^3-2)/(x^2+x)>=0;
```

$$olikhet := 0 \le \frac{x^3 - 2}{x^2 + x}$$

```
> solve(olikhet);solve(olikhet,{x});
```

$$\text{RealRange}(\text{Open}(-1), \text{Open}(0)), \text{RealRange}(2^{(1/3)}, \infty)$$

$$\{-1 < x, x < 0\}, \{2^{(1/3)} \le x\}$$

2. Funktioner, gränsvärde, derivering, plot

maple använder samma beteckningar för funktioner som vi. Skilja alltid mycket noggrant mellan "*expression*" (uttryck) som t.ex. $f := 2\sin\left(\frac{x}{2}\right) + \cos(3\,x)$ och "*procedure*" (funktioner på operatorform), t.ex. $g := x \to 2\sin\left(\frac{x}{2}\right) + \cos(3\,x)$. Det sista (den "elementvisa tillordningen", så som även vi skriver) är vida att föredra. Vill du beräkna vår funktion för t.ex. $x = \pi$ så sätter du helt enkelt in π i funktionen *g*, medan du måste ersätta ("substituera") x med π i uttrycket f och räkna ut detta sedan med *simplify* (förenkla):

Exempel 9

```
> f:=2*sin(x/2)+cos(3*x); g:=x->2*sin(x/2)+cos(3*x);
```

$$f := 2\sin\left(\frac{x}{2}\right) + \cos(3\,x)$$

$$g := x \to 2\sin\left(\frac{1}{2}\,x\right) + \cos(3\,x)$$

```
> g(Pi); subs(x=Pi,f); simplify(%);
```

$$1$$

$$2\sin\left(\frac{\pi}{2}\right) + \cos(3\,\pi)$$

$$1$$

Derivatan får du med deriveringsoperatorn D för funktioner (som funktion), resp. med *diff...* för uttryck (som uttryck):

```
> D(g); D(g)(x); diff(f,x);
```

$$x \to \cos\left(\frac{1}{2}\,x\right) - 3\sin(3\,x)$$

$$\cos\left(\frac{x}{2}\right) - 3\sin(3\,x)$$

$$\cos\left(\frac{x}{2}\right) - 3\sin(3\,x)$$

Sammansättningen av funktioner (och operatorer) fås med @:

Exempel 10

```
> f:=sin; g:=sqrt; (f@g)(x);(g@f)(x);
```

$$f := \sin$$

$$g := \mathrm{sqrt}$$

$$\sin(\sqrt{x})$$

$$\sqrt{\sin(x)}$$

Användingen av samma operator n gånger fås med @@n; t.ex. fås den andra derivatan $f'' =$ $\mathrm{D}(\mathrm{D}(f)) = (\mathrm{D@D})(f)$ =(D@@2)(f). Kolla följande exempel:

Exempel 11

```
> g:=x->x^3; (g@g)(x); (g@@2)(x); (g@@3)(x); (D@@2)(g@g)(x);
```

$$g := x \to x^3$$

$$x^9$$

$$x^9$$

$$x^{27}$$

$$72\, x^7$$

Deriverar du ett uttryck med *diff* n gånger, så skriver du n gånger x eller x\$$n$:

```
> diff((g@g)(x),x,x); diff((g@g)(x),x$2);
```

$$72\, x^7$$

$$72\, x^7$$

Gränsvärdet beräknar du med *limit* (=limes), ensididiga med tillägget *left* resp. *right*:

Exempel 12

```
> limit(sin(a*x)/x,x=0); limit(sin(a*x)/x,x=infinity);
```

$$a$$

$$0$$

```
> limit(x/abs(x),x=0,'left'); limit(x/abs(x),x=0,'right');
limit(x/abs(x),x=0);
```

$$-1$$

$$1$$

$$\mathit{undefined}$$

Ritar funktioner gör du med *plot*. För att rita en kurva beräknas funktionsvärdena i punkter x_k och dras sträckor mellan punkterna (x_k, $\mathrm{f}(x_k)$) och (x_{k+1}, $\mathrm{f}(x_{k+1})$); punkterna x_k väljes smart, men du kan öka antalet med *numpoints*. Där funktionen har ett språng, kan du förhindra att *maple* ritar sträckor med *discont* = *true*. Det funkar dock inte alltid, t.ex. inte om f inehåller belopp. Vi ritar som exempel arcustangens:

<u>Exempel 13</u>

```
> plot([arctan(1/x),0],x=-8..8,numpoints=999,discont=true);
```



Med *discont* får du de punkter, i vilka f ej är kontinuerlig:

```
> discont(arctan(1/x),x);
```

$$\{0\}$$

Vi genomför nu två typtal:

<u>Exempel 14</u>

Låt $f(x)=\frac{\ln(\cos(x))}{x}$. Har f ett gränsvärde då x går mot 0? Om ja, blir f deriverbar i 0 om vi definierar funktionsvärdet $f(0)$ som detta gränsvärde? Är f konvex eller konkav?

Lösning:
Observera det enklare sättet då f skrives som procedure! Vi ritar även x-axeln; tänk på definitionsmängden!

```
> f:=x->ln(cos(x))/x^2; limit(f(x),x=0);
```

$$f := x \to \frac{\ln(\cos(x))}{x^2}$$

$$\frac{-1}{2}$$

```
> plot({f(x),0},x=-Pi/2..Pi/2,y=-2..0,numpoints=999,color=red):
```
eller
```
plot({f,0},-Pi/2..Pi/2,y=-2..0,numpoints=999,color=red);
```



Vi sätter alltså $f(0) = -\frac{1}{2}$ och kollar om f blir då deriverbar i 0, dvs. om differenskvoten har ett gränsvärde då x går mot 0 (vi använder samma symbol f för denna nya funktion):

```
> diffkvot:=(f(x)+1/2)/x; limit(diffkvot,x=0);
```

$$diffkvot := \frac{\frac{\ln(\cos(x))}{x^2} + \frac{1}{2}}{x}$$

$$0$$

Detta gränsvärde räknar vi ut i del C! Det visar att f är deriverbar i 0 med $D(f)(0) = 0$; kolla dock med höger- vänstergränsvärde (räkna ut bägge, eller jämför direkt med boolesk evaluering):

```
>
evalb(limit(diffkvot,x=0,'left')=limit(diffkvot,x=0,'right'
));
```

$$true$$

Här var detta dock onödigt ty f är jämn:

```
> type(f(x),evenfunc(x));
```

$$true$$

Kolla om derivatan är kontinuerlig i 0:

```
> fprim:=D(f);
```

$$fprim := x \to -\frac{\sin(x)}{\cos(x)\, x^2} - \frac{2\ln(\cos(x))}{x^3}$$

```
> limit(fprim(x),x=0);
```

$$0$$

Det visar att derivatan är kontinuerlig i 0. Vi ritar derivatan och ser att den är avtagande, alltså att f är konkav. Detta fås (bättre) genom att rita andraderivatan, som visar sig vara <0 (kolla dock först om den existerar i 0):

```
> plot(fprim(x),x=-Pi/2..Pi/2,y=-4..4); plot((D@D)(f)(x),x=-
Pi/2..Pi/2,y=-3..0);
```





Exempel 15

Detta var uppgift 4b) på tentan 97-08-22 (lös den först helt för hand!):

Låt $\mathrm{f}(x) = (1 - x^2)\,\mathbf{e}^{\left(-\frac{x^2}{2}\right)}$. Rita kurvan $y = \mathrm{f}(x)$ med angivande av extrempunkter, asymptoter och konvexitet/konkavitet.

Lösning:

För att kunna rita fler figurer (med *display*) laddar vi in *plots*-paketet och för att kunna rita enstaka punkter ladder vi in *plottools*-paketet:

```
> with(plots):with(plottools):
```

Nu beräknar vi de stationära punkterna och var f är växande ($f'>0$ resp. <0); observera igen hur noggrant *maple* anger intervallen:

```
> f:=x->(1-x^2)*exp(-x^2/2);
```

$$f := x \to (1 - x^2)\,\mathbf{e}^{(-1/2\,x^2)}$$

```
> solve(D(f)(x),x);solve(D(f)(x)<0);solve(D(f)(x)>0);
```

$$0, \sqrt{3}, -\sqrt{3}$$

$$\mathrm{RealRange}(-\infty, \mathrm{Open}(-\sqrt{3})), \mathrm{RealRange}(\mathrm{Open}(0), \mathrm{Open}(\sqrt{3}))$$

$$\mathrm{RealRange}(\mathrm{Open}(-\sqrt{3}), \mathrm{Open}(0)), \mathrm{RealRange}(\mathrm{Open}(\sqrt{3}), \infty)$$

Sak samma för att få inflexionspunkterna (och var f är konvex; analogt konkav); om du anger m.a.p. vilken variabel ekvationen/olikheten skall lösas svarar *maple* ändå mera lättläst:

```
> solve((D@D)(f)(x),{x});solve((D@D)(f)(x)>0,{x});
```

$$\{x = -\sqrt{3+\sqrt{6}}\}, \{x = \sqrt{3+\sqrt{6}}\}, \{x = -\sqrt{3-\sqrt{6}}\}, \{x = \sqrt{3-\sqrt{6}}\}$$

$$\{-\sqrt{3+\sqrt{6}} < x, x < -\sqrt{3-\sqrt{6}}\}, \{\sqrt{3-\sqrt{6}} < x, x < \sqrt{3+\sqrt{6}}\}$$

Nu skapar vi de olika plots: punkterna ritas med *point* (jag valde små cirklar med *symbol=CIRCLE*), sedan ritas alla i samma figur med *display*. Vi ritar bara för $x>0$ (f är ju jämn!):

```
> type(f(x),evenfunc(x));
```

$$true$$

```
> p1:=plot(f,0..4,numpoints=880,labels=[x,y]):
>
p2:=point([sqrt(3),f(sqrt(3))]):p3:=point([1,0]):p4:=point(
[(3+6^(1/2))^(1/2),f((3+6^(1/2))^(1/2))]):p5:=point([(3-
6^(1/2))^(1/2),f((3-6^(1/2))^(1/2))]):
> display([p1,p2,p3,p4,p5,plot(0,x=0..4)],symbol=CIRCLE);
```




```
> limit(f(x),x=infinity);
```

$$0$$

Detta visar att x-axeln är asymptot.

3. Integral

Först beräknar vi en primitiv funktion F till funktionen f och plottar F:

Exempel 16

```
> f:=x->(-6*x^5+4*x^4+18*x^3-17*x^2+9)/(x^6-4*x^4+3*x^2):
```

Partialbråksuppdelning fås enkelt med (omvandla till *partial fraction*):

```
> convert(f(x),parfrac,x);
```

$$-\frac{2}{x-1}-\frac{4}{x+1}+\frac{3}{x^2}-\frac{1}{-3+x^2}$$

Ser du att nämnaren har roten $\sqrt{3}$, kan du skriva (men för att integrera med *maple* behövs det ej):

```
> convert(f(x),parfrac,x,3^(1/2));
```

$$-\frac{2}{x-1}-\frac{4}{x+1}-\frac{\sqrt{3}}{6\,(x-\sqrt{3})}+\frac{3}{x^2}+\frac{\sqrt{3}}{6\,(x+\sqrt{3})}$$

```
> F:=int(f(x),x);
```

$$F := -\frac{3}{x}-2\ln(x-1)-4\ln(x+1)+\frac{1}{3}\sqrt{3}\,\mathrm{arctanh}\left(\frac{x\sqrt{3}}{3}\right)$$

För att kunna plotta F, skriver vi om arctanh m.h.a. ln (med "*convert*") och ersätter sedan i F arctanh med den så erhållna funktionen:

```
>
F1:=subs(arctanh(1/3*x*sqrt(3))=convert(arctanh(1/3*x*sqrt(3)),ln),F);
```

$$F1 := -\frac{3}{x} - 2\ln(x-1) - 4\ln(x+1) + \frac{1}{3}\sqrt{3}\left(\frac{1}{2}\ln\left(\frac{x\sqrt{3}}{3}+1\right) - \frac{1}{2}\ln\left(1-\frac{x\sqrt{3}}{3}\right)\right)$$

Till sist ersätter vi $\ln(x)$ med $\ln(|x|)$:

```
> F2:=simplify(subs(ln=ln@abs,F1));
```

$$F2 := -\frac{1}{6}\,\frac{18 + 12\ln(|x-1|)\,x + 24\ln(|x+1|)\,x - x\sqrt{3}\,\ln(|x+\sqrt{3}|) + x\sqrt{3}\,\ln(|x-\sqrt{3}|)}{x}$$

Då kan vi nu plotta över ett godt. intervall. Eftersom f ej är kontinuerlig, så plottar vi med *discont=true* (se ex13) och begränsar *y*-värdena som skall vara med:

```
> plot([F2,0],x=-2..2,y=-13..20,color=red,numpoints=1234,discont=true,title=`kurvan y=F(x)`);
```



Nu till **Riemannsummor**: vi tar som exempel funktionen $2\,\operatorname{arcsinh}(x)\sin(x) + \frac{x}{2} - \frac{\pi}{4}$

:

__Exempel 17__

Vänster-, mitt-, höger-Riemannsummor kan vi rita m.h.a. *leftbox*..., som finns i *student*-paketet:

```
> with(student):
```

> $g := x \to 2\,\operatorname{arcsinh}(x)\sin(x) + \frac{x}{2} - \frac{\pi}{4}$

```
> left:=leftbox(g(x),x=-3..Pi,33,color=red,thickness=2,shading=yellow):
> mitt:=middlebox(g(x),x=-3..Pi,33,color=red,thickness=2,shading=green):
> right:=rightbox(g(x),x=-3..Pi,33,color=red,thickness=2,shading=cyan):
```

Rita nu en efter en, eller alla i samma plot m.h.a. *display* (skriv in dem som en lista):

```
> with(plots):
display([left,mitt,right],title=`Riemannsummor (n=33)`);
```



Summorna kan förstås beräknas exakt:

> `middlesum(g(x),x=-3..Pi,33):` men vi vill ju ha ett närmevärde:

> `evalf(leftsum(g(x),x=-3..Pi,33));evalf(middlesum(g(x),x=-3..Pi,33));evalf(rightsum(g(x),x=-3..Pi,33));`

4.500496250

4.769841900

4.976480469

Jämför med närmevärdet (med 9 korrekta decimaler)

> `evalf(int(g(x),x=-3..Pi));`

4.759387315

Lycka till! Bernhard, oktober 2004